

STA442S 2008 Assignment 6, Question 4

Main question - whether crustiness of bread depends on temperature

- $F = 7.21$
- p-value = 0.0715
- No conclusion as the result is not statistically significant at the 0.05 level.
- Since the effect of temperature is not significant, follow up Scheffé tests are not necessary.

Secondary question - whether there is any difference in crustiness from batch to batch

- $F = 6.26$
- p-value = 0.0084
- The result is statistically significant at the 0.05 level, average crustiness of bread differs between batches for at least one of the temperature levels.
- Follow up tests are not required on significant effect for factor with only two levels. This is because if the effect is significant, we know the difference comes from level 1 and level 2.
- However, it is not the case for nested factor. Batch 1 for low temperature is different from batch 1 for medium temperature. A significant batch effect (nested effect) simply tells us mean crustiness differs between batches for *at least* one of the temperature levels.
- Low temperature, batch 1 vs 2
 - $F = 12.51$
 - Critical value = 10.470884
 - The result is statistically significant at the 0.05 level, breads from batch 2 is more crusty than those from batch 1 on average with low temperature settings.
- Medium temperature, batch 1 vs 2
 - $F = 3.13$
 - Critical value = 10.470884
 - No conclusion as the result is not statistically significant at the 0.05 level.
- High temperature, batch 1 vs 2
 - $F = 3.13$
 - Critical value = 10.470884

– No conclusion as the result is not statistically significant at the 0.05 level.

```
/****** bread.sas *****/

title 'Bread:Two-factor mixed nested model with proc glm';

options linesize=79 noovp formdlim='_';

data bread;
  infile 'bread.data';
  input temp batch crust;

label
  temp = 'Temperature (1-Low, 2-Medium, 3-High)'
  batch = 'Batch Number'
  crust = 'Crustiness of Bread';

proc glm;
  title2 'Temperature fixed, batch random';
  title3 'Temperature (1-Low, 2-Medium, 3-High)';
  class temp batch;
  model crust = temp batch(temp);
  random batch(temp) / test;
  means temp batch(temp);

  /* Follow up tests for nested effect of batch within temperature.
  We do this because the nested effect was significant. */
  contrast 'Batch 1 vs 2, Low Temperature'
    batch(temp) 1 -1 0 0 0 0;
  contrast 'Batch 1 vs 2, Medium Temperature'
    batch(temp) 0 0 1 -1 0 0;
  contrast 'Batch 1 vs 2, High Temperature'
    batch(temp) 0 0 0 0 1 -1;

  /* Follow up tests for effect of temperature.
  This is done to show you how to set up the contrasts.
  Note that we have to specify the error term.
  To know if you need to specify the error term, run the
  proc glm without the contrasts. The output will show. */
  contrast 'Low vs Medium'
    temp 1 -1 0 / e = batch(temp);
  contrast 'Low vs High'
    temp 1 0 -1 / e = batch(temp);
  contrast 'Medium vs High'
    temp 0 1 -1 / e = batch(temp);
```

```

/* For follow up tests for nested effect of batch within
temperature, we consider the initial test to be test for the nested
effect, rather than the overall F test. */ proc iml;
    title2 'Follow up tests for nested effect of batch within temperature';
    title3 'Table of critical values for all possible Scheffe tests';
    numdf = 3; /* Numerator degrees of freedom for initial test */
    dendf = 12; /* Denominator degrees of freedom for initial test */
    alpha = 0.05;
    critval = finv(1-alpha,numdf,dendf);
    zero = {0 0}; S_table = repeat(zero,numdf,1); /* Make empty matrix */
    /* Label the columns */
    namz = {"Number of Contrasts in followup test"
    " Scheffe Critical Value"}; mattrib S_table colname=namz;
    do i = 1 to numdf;
        s_table(|i,1|) = i;
        s_table(|i,2|) = numdf/i * critval;
    end;
    reset noname; /* Makes output look nicer in this case */
    print "Initial test has" numdf " and " dendf "degrees of freedom."
    "Using significance level alpha = " alpha;
    print s_table;

```

```

/* For follow up tests for main effect of temperature, we consider
the initial test to be test for the main effect, rather than the
overall F test. */ proc iml;
    title2 'Follow up tests for main effect of temperature';
    title3 'Table of critical values for all possible Scheffe tests';
    numdf = 2; /* Numerator degrees of freedom for initial test */
    dendf = 3; /* Denominator degrees of freedom for initial test */
    alpha = 0.05;
    critval = finv(1-alpha,numdf,dendf);
    zero = {0 0}; S_table = repeat(zero,numdf,1); /* Make empty matrix */
    /* Label the columns */
    namz = {"Number of Contrasts in followup test"
    " Scheffe Critical Value"}; mattrib S_table colname=namz;
    do i = 1 to numdf;
        s_table(|i,1|) = i;
        s_table(|i,2|) = numdf/i * critval;
    end;
    reset noname; /* Makes output look nicer in this case */
    print "Initial test has" numdf " and " dendf "degrees of freedom."
    "Using significance level alpha = " alpha;
    print s_table;

```